
Abstract. The aim of this paper is to describe a Stata routine for the nonparamet-
ric estimation of mixed logit models with an Expectation-Maximisation algorithm
proposed in Train (2008). We also show how to use the Stata command lclogit,
which performs the estimation automatically.

1 Introduction

In a recent contribution Train (2008) has showed how EM algorithms can be used for
the nonparametric estimation of mixing distributions in discrete choice models. In this
paper we consider one of the three nonparametric methods he proposes and show how
it can be implemented in Stata. The method presented here allows estimating discrete
mixing distributions with mass points and share probabilities as parameters. Therefore,
the nonparametric estimation is based on a typical latent class model and is reached
by increasing the number of mass points of each coefficient so as to approximate their
mixing distributions.

Traditionally, latent class models have been estimated using gradient-based algo-
rithms, such as Newton-Raphson or BHHH. However, the estimation based on standard
optimization procedures becomes difficult when the number of mass points increases, as
the higher the number of latent classes the more difficult the empirical inversion of the
Hessian matrix, with the possibility of singularity at some iterations. In such situation
an EM algorithm could help as it implies the repeated evaluation of a function that is
far easier to maximize. Moreover, EM recursions have proved to be particularly stable
and - under conditions given by Boyles (1983) and Wu (1983) - they always climb
uphill until convergence to a local maximum.

The paper is structured as follows: section 2 presents a mixed logit model based on
discrete mixed distributions; section 3 shows how this model can be estimated via EM
algorithm; section 4 presents a detailed step-by-step description of EM estimation in
Stata; section 5 introduces the Stata command lclogit; section 6 contains an empirical
application based on accessible data; section 7 concludes.

c© yyyy StataCorp LP st0001



2 An EM algorithm for nonparametric mixed logit models

2 A mixed logit model with discrete mixing distributions

Assume there are N agents who face J alternatives on T choice occasions. Agents
choose the alternative that maximizes their utility in each choice occasion. The random
utility of agent i from choosing alternative j in period t is defined as follows:

Uijt=βixijt + εijt (1)

Where xijt is a vector of alternative-specific attributes and εijt is a stochastic term,
which is assumed to be distributed IID extreme value. Importantly, each βi is assumed
to be random with unconditional density f(β |ϑ), where ϑ collects the parameters that
characterize the distribution.

Conditional on knowing βi the probability of the observed sequence of choices for
agent i is given by the traditional McFadden’s choice model:1

Pri(β) =

T∏
t=1

J∏
j=1

(
exp(βixijt)∑J

k=1 exp(βixikt))

)dijt
(2)

Where dijt is a dummy that selects the chosen alternative in each choice occasion.
However, since βi is unknown the conditional probability of the sequence of observed
choices has to be evaluated for any possible value of βi. Hence, assuming that f(β |ϑ)
has a continuous distribution, the unconditional probability becomes:

Pri(ϑ) =

∫ T∏
t=1

J∏
j=1

(
exp(βxijt)∑J

k=1 exp(βxikt))

)dijt
f(β|ϑ) (3)

Typically, the log likelihood function derived from this model is estimated with simula-
tion methods.2

If the distribution of each βi is discrete, the probability in equation 3 becomes:

Pri(ϑ) =

C∑
c=1

πc

T∏
t=1

J∏
j=1

(
exp(βcxijt)∑J

k=1 exp(βcxikt))

)dijt
(4)

Where πc = f(βc|ϑ) represents the share of the population with coefficients βc.

Equation 4 is a typical latent class model. Nevertheless, here we follow the classifica-
tion proposed by McFadden and Train (2000) and define it as a discrete mixed model,
in order to emphasize the similarities with the continuous mixed model of equation 3.

The estimation of discrete mixed models is usually based on standard gradient-based
methods. However, these methods often fail to achieve convergence when the number of
latent classes becomes high. In this case an EM algorithm could help, as it requires the
repeated maximization of a function that is far simpler with respect to the log likelihood
derived from equation 4.

1. See McFadden (1973).
2. See Train (2003). In Stata, continuous mixed logit models can be estimated with Simulated

Maximum Likelihood through the program mixlogit, written by Hole (2007).

3 An EM algorithm for the estimation of mixed logit
models with discrete mixing distributions

EM algorithms were initially proposed in the literature to deal with missing data prob-
lems. Nevertheless, they turned out to be an efficient method to estimate latent class
models, where the missing information consists of the class share probabilities. Nowa-
days, they are widely used in many economic fields where the assumption that people
can be grouped in classes with different unobserved preference heterogeneity is reason-
able.

The recursion is known as “E-M” because it consists of two steps, namely an “Expec-
tation” and a “Maximization”. As explained in Train (2008), the term to be maximized
is the expectation of the missing-data log likelihood - i.e the joint density of the ob-
served choice and the missing data - whilst the expectation is over the distribution of the
missing information, conditional on the density of the data and the previous parameter
estimates.

Consider the conditional logit model with discrete mixing distributions outlined in
the previous section. Following equation 4, the log likelihood is defined as:

LL =

N∑
i=1

ln

C∑
c=1

πc

T∏
t=1

J∏
j=1

(
exp(βcxijt)∑J

k=1 exp(βcxikt))

)dijt
(5)

Which can be maximized by means of standard, gradient-based optimization methods.
However, the same log likelihood can be also maximized by repeatedly updating the
following recursion:

ηs+1 = argmaxη
∑
i

∑
c Ci(η

s)ln πc
∏
t

∏
j

(
exp(βcxijt)∑J

k=1 exp(βcxikt))

)dijt
= argmaxη

∑
i

∑
c Ci(η

s)ln(Li | classi = c)

(6)

Where η is as vector that contains the whole set of parameters to be estimated - i.e.
those that enter the probability of the observed choice plus those that may define the
class shares - Li is the missing-data likelihood function and Ci(η

s) is the conditional
probability that household i belongs to class c, which depends on the density of the
data and the previous value of the parameters.

This conditional probability - Ci(η
s) - is the key future of the EM recursion and can

be computed by means of the Bayes’ theorem:

Ci(η
s) =

Li|classi = c∑C
c=1 Li|classi = c

(7)

Now, given that:

ln(Li | classi = c) = ln πc + ln

T∏
t=1

J∏
j=1

(
exp(βcxijt)∑J

k=1 exp(βcxikt))

)dijt
(8)

The recursion in equation 6 can be split into the following steps:

1. Form the contribution to the likelihood (Li | classi = c) as defined in equation 6
for each class;3

2. Form the individual-specific conditional probabilities of class membership as in
equation 7;

3. Following equation 8, update the sets of βc, c = 1, 2, ..., C by maximizing - for
each class - a conditional logit model with weighed observations, with weights
given by the conditional probabilities of class membership:

βs+1
c = argmaxβ

N∑
i=1

C(ηs)ln

T∏
t=1

J∏
j=1

(
exp(βcxijt)∑J

k=1 exp(βcxikt))

)dijt
(9)

4. Maximize the other part of the log likelihood in equation 6 and get the updated
vector of class shares:

πs+1 = argmaxπ

N∑
i=1

C∑
c=1

Ci(η
s)ln πc (10)

• If the class share probabilities depend on a vector of demographics - zi - the
relative parameters are updated as:

αs+1 = argmaxα

N∑
i=1

C∑
c=1

Ci(η
s)ln

exp(αczi)∑
c exp(αczi)

, αC = 0 (11)

This is a grouped-data log likelihood, where we have used a logit specification
so as to constrain the estimated class share probabilities into the right range.4

The updated class share probabilities - πc, c = 1, 2, ..., C - are then computed
as:

πs+1
c =

exp(α̂s+1
c zi)∑

c exp(α̂
s+1
c zi)

, c = 1, 2, ..., C (12)

Which, in turn, allows updating the conditional probabilities of class mem-
bership by means of the new vectors βs+1

c and πs+1
c , c = 1, 2, ..., C.

• If the class share probabilities do not depend on demographics the empirical
maximization of the function in equation 11 can be avoided, as its analytical
solution would be given by:

πs+1
c =

∑
i Ci(η

s+1)∑
i

∑
c Ci(η

s+1)
, c = 1, ..., C (13)

3. For the first iteration, starting values have to be used for the densities that enter the model. Note
that these starting values must be different in every class. Otherwise, the recursion estimates the
same set of parameters for all the classes.

4. Differently from the βcs, the vectors αc, c = 1, 2, ..., C are jointly estimated. This is needed in
order to ensure that

∑
c πc = 1.

Where the updated conditional probabilities - Ci(η
s+1) - are computed by

using the updated values of βc, c=1,2,...,C and the previous values of the
class shares.

5. Once the conditional probabilities of class membership have been updated - either
in models with or without covariates on zi - the recursion can start again from
point 3 until convergence.

4 A Stata routine for the estimation of mixed logit mod-
els with discrete mixing distributions

In this section we show how the EM algorithm outlined above can be coded into Stata.
We propose a detailed step-by-step procedure that can be easily implemented in a simple
do file and work with accessible data from Huber and Train (2000) on household’s choice
of electricity supplier. Note that this is the same database used by Hole (2007) for an
application of his Stata program mixlogit, which performs the estimation of parametric
mixed logit models via Simulated Maximum Likelihood.

The data collects information on 100 residential electricity customers, who were
asked up to 12 choice experiments.5 In each experiment the customer was asked which
of the 4 suppliers he/she would prefer among four hypothetical electricity suppliers.

The following characteristics of each offer were stated:

• The price of the contract (in cents per kWh) whenever the supplier offers a contract
with a fixed rate (price)

• The length of contract that the supplier offered, expressed in years (contract)

• Whether the supplier is a local company (local)

• Whether the supplier is a well-known company (wknown)

• Whether the supplier offers a time-of-day rate instead of a fixed rate (tod)

• Whether the supplier offers a seasonal rate instead of a fixed rate (seasonal)

Each costumer is identified by the variable pid. For each costumer, the variable gid

identifies a given choice occasion, while the dummy variable y identifies the stated choice
in each choice occasion.

The data setup required for estimating the model is as follows:

. use http://fmwww.bc.edu/repec/bocode/t/traindata.dta, clear

. list in 1/12, sepby(gid)

5. Since some customers stopped before answering all 12 experiments, there is a total of 1195 choice
occasions in the sample.

y price contract local wknown tod seasonal gid pid

1. 0 7 5 0 1 0 0 1 1
2. 0 9 1 1 0 0 0 1 1
3. 0 0 0 0 0 0 1 1 1
4. 1 0 5 0 1 1 0 1 1

5. 0 7 0 0 1 0 0 2 1
6. 0 9 5 0 1 0 0 2 1
7. 1 0 1 1 0 1 0 2 1
8. 0 0 5 0 0 0 1 2 1

9. 0 9 5 0 0 0 0 3 1
10. 0 7 1 0 1 0 0 3 1
11. 0 0 0 0 1 1 0 3 1
12. 1 0 0 1 0 0 1 3 1

The next subsection presents the steps for estimating a model with covariates on the
class share porbabilities. Alternatively, the optimization of the algorithm when only a
constant term is included among the class probabilities is presented in subsection 4.2.

4.1 A model with covariates on the class share probabilities

In order to present a flexible routine we work with global variables, so that the code
can be easily adapted to other databases. The dependent variable is called $depvar,
the list of covariates that enter the probability of the observed choice $varlist; the
list of variables that enter the grouped-data log likelihood $varlist2; the variable that
identifies the panel dimension - i.e. the choice makers - $id; the variable that defines
the choice situations for each choice maker $group. We also define the number of latent
classes $nclasses and the number of maximum iterations $niter.6

. **(1) Set the estimation setup**

. global depvar "y"

. global varlist "price contract local wknown tod seasonal"

. gen _con=1

. global varlist2 "_con x2"

. global id "pid"

. global group "gid"

. global nclasses "2"

. global niter "35"

In order to compute the starting values, we randomly split the sample into C different
sub-samples - one for each class - and estimate a separate clogit for each of them.7

6. In the following estimation setup the covariate that we included in $varlist2 - x2 - can be created
manually, as the database does not contain individual-level variables.

7. If the same starting values were used for all the classes the EM algorithm would perform the same
computations for each class and return the same results at each iteration.

After each clogit estimation we use the command predict to obtain the probability
of every alternative in each class and we store them in the variables l1, l2,..., lC.

As for the starting values for the probability of class membership we simply define
equal shares, that is 1

C :

. **(2) Split the sample**

. bysort $id: gen double _p=runiform() if _n==_N

. bysort $id: egen double _pr=sum(_p)

. local prop 1/$nclasses

. gen double _ss=1 if _pr<=`prop´

. forvalues s=2/$nclasses {

. replace _ss=`s´ if _pr>(`s´-1)*`prop´ & _pr<=`s´*`prop´

. }

. **(3) Get starting values for the beta coefficients and the class shares**

. forvalues s=1/$nclasses {

. gen double _prob`s´=1/$nclasses

. clogit $depvar $varlist if _ss==`s´, group($group) technique(nr dfp)

. predict double _l`s´

. }

In what follows, the steps to calculate the conditional probabilities of equation 7
from these starting values are presented.

First, for each latent class we multiply the variables l1, l2,..., lC by the dummy
variable that identifies the observed choice in each choice situation, i.e. $depvar. Note
that this allows storing the probabilities of the observed choice for each class.

Second, for each latent class we multiply the probabilities of the observed choices
in each choice situation in order to obtain the probability of the agent’s sequence of
choices:8

. **(4) Compute the probability of the agent´s sequence of choices**

. forvalues s=1/$nclasses {

. gen double _kbb`s´=_l`s´*$depvar

. recode _kbb`s´ 0=.

. bysort $id: egen double _kbbb`s´=prod(_kbb`s´)

. bysort $id: replace _kbbb`s´=. if _n!=_N

. }

Third, we construct the denominator of equation 7, i.e. the unconditional choice
probability for each choice maker. This is done by computing a weighted average of the
probabilities of the agent’s sequence of choices in each class, with weights given by the
class shares, i.e. the variables prob1, prob2,..., probC:

. **(5) Compute the choice probability**

8. This last step is done through the user-written program gprod. Type findit gprod and install
the package dm71.

. gen double _den=_prob1*_kbbb1

. forvalues s=2/$nclasses {

. replace _den=_den+_prob`s´*_kbbb`s´

. }

Finally, we compute the ratios defined in equation 7 and rearrange them in order
to create individual-level variables. These are the conditional probabilities of class
membership as defined in the previous section:

. **(6) Compute the conditional probabilities of class membership**

. forvalues s=1/$nclasses {

. gen double _h`s´=(_prob`s´*_kbbb`s´)/_den

. bysort $id: egen double _H`s´=sum(_h`s´)

. }

Before starting the loop that iterates the EM recursion until convergence, we need
to specify a Stata ml program that performs the estimation of the grouped-data model
defined in equation 11:9

**(7) Provide Stata with the ML command for the grouped-data model**

. bysort $group: gen _alt=sum(1)

. su _alt

. bysort $id: gen double _id1=1 if _n<=r(mean)

. program logit_lf

. args lnf a2 a3 a4 a5 a6 a7 a8 a9 a10 a11 a12 a13 a14 a15 a16 a17 a18 a19 a20

. tempvar denom

. gen double `denom´=1

. forvalues c=2/$nclasses {

. replace `denom´=`denom´+exp(`a`c´´)

. }

. replace `lnf´=_H1*ln(1/`denom´) if $depvar==1 & _id1==1

. forvalues c=2/$nclasses {

. replace `lnf´=`lnf´+_H`c´*ln(exp(`a`c´´)/`denom´) if $depvar==1 & _id1==1

. }

. replace `lnf´=0 if `lnf´==.

. **Note: the ML command updates the class shares internally:**

. capture drop _prob*

. gen double _prob1=1/`denom´

. forvalues c=2/$nclasses {

. gen double _prob`c´=exp(`a`c´´)/(`denom´)

. }

. end

There are two important remarks about the above-mentioned routine. First, we set
to zero one vector of parameters for identification. Second, the class shares - prob1,

9. The ml program presented in the following lines allows for up to 20 latent classes. However, the
routine can be easily modified so as to account for more - or less - classes.

prob2,..., probC - are updated internally when the ml program is called.

We now present the loop that repeats the steps above until convergence:

. local i=1

. while `i´<= $niter{

To begin with, estimate again the C clogit models (one for each class) using the
conditional probabilities of class membership as weights. Then, recompute the proba-
bility of the chosen alternative in each choice occasion so as to update probabilities of
the agent’s sequence of choices:

**(8) Update the probability of the agent´s sequence of choices**
. capture drop _l* _kbbb*

. forvalues s=1/$nclasses {

. clogit $depvar $varlist [iw=_H`s´], group($group) technique(nr dfp)

. predict double _l`s´

. replace _kbb`s´=_l`s´*$depvar

. recode _kbb`s´ 0=.

. bysort $id: egen double _kbbb`s´=prod(_kbb`s´)

. bysort $id: replace _kbbb`s´=. if _n!=_N

. }

Now launch the ml model for the estimation of the grouped-data log likelihood:

**(9) Update the class share probability**

. global classes="($varlist2)"

. forvalues s=3/$nclasses {

. global classes="$classes ($varlist2)"

. }

. ml model lf logit_lf $classes, max

Once the class share probabilities have been updated within the ml program, the
next step requires updating both the choice probability - i.e. the variable den - and
the conditional probabilities of class membership, i.e. the variables H*:

**(10) Update the choice probability**

. replace _den=_prob1*_kbbb1

. forvalues s=2/$nclasses {

. replace _den=_den+_prob`s´*_kbbb`s´

. }
**(11) Update the conditional probabilities of class membership**

. drop _H*

. forvalues s=1/$nclasses {

. replace _h`s´=(_prob`s´*_kbbb`s´)/_den

. bysort $id: egen double _H`s´=sum(_h`s´)

. }

Once the parameters and the conditional probabilities have been updated the routine
computes the log likelihood and restarts the loop until convergence.

As pointed out in Train (2008), convergence of EM algorithms for nonparametric
estimation is still controversial and constitutes an area of future research. As in Train
(2008) and Weeks and Lange (1989) here we define convergence when the change in the
log likelihood function from one iteration to the other is sufficiently small. Hence, the
routine automatically stops the internal loop before the selected number of iterations
provided that the log likelihood has not changed more than 0.001% during the last five
iterations:10

. **(12) Update the log likelihood**

. capture drop _sumll

. egen _sumll=sum(ln(_den))

. sum _sumll

**Check if the log likelihood has increased from the previous 5 iterations:**
. global z=r(mean)

. local _sl`i´=$z

. if `i´>=6 {

. local a=`i´-5

. }
**Stop the loop if the LL has not change of more than 0.001% over the last 5 iterations:**

. if `i´>=6 {

. local `_vpsl`i´´= -(`_sl`i´´ - `_sl`a´´)/`_sl`a´´

. if `_vpsl`i´´<= 0.00001 {

. local i=$niter

. }

. }
**Restart the loop and display the log likelihood**
. local i=`i´ +1

. display as green "Iteration " `i´ ": log likelihood = " as yellow $z

. }

Once the algorithm has converged the results can be displayed by typing:

. forvalues s=1/$nclasses {

. clogit $depvar $varlist [iw=_H`s´], group($group) technique(nr dfp)

. }

4.2 A model without covariates on the class shares probabilities

When the model does not include demographics on the class share probabilities the
maximization of the grouped-data log likelihood can be avoided. In fact, its solution
can be provided analytically as it is shown in equation 13. This is important because
the maximization of the grouped-data model slows down the overall estimation process,
which could become time-consuming when the number of latent classes is high.

If there are no covariates on the class share probabilities the EM algorithm can be
optimized by simply dropping out the 9th step from the loop presented in the previous

10. Such a small value is advisable because - as discussed in Dempster et al. (1977) - EM algorithms
may move very slowly when they are close to a maximum.

subsection. Therefore - following the routine - the choice probability is updated by
means of the previous values of the class shares and this, in turn, allows updating the
conditional probabilities of class membership during the 11th step.

Once the conditional probabilities have been updated they are used to construct the
numerator and the denominator of equation 13, so that also the class share probabilities
can be updated to the next iteration. Hence, the following lines should be added after
the 11th step:

**Compute the numerator of equation 13**

. forvalues s=1/$nclasses {

. capture drop _nums`s´

. egen double _nums`s´=sum(_h`s´)

. }
**Compute the denominator of equation 13**

. capture drop _dens

. gen double _dens=_nums1

. forvalues s=2/$nclasses {

. replace _dens=_dens+_nums`s´

. }

. **Update the class shares**

. forvalues s=1/$nclasses {

. replace prob`s´=nums`s´/dens

. }

Subsequently, the loop continues as before from point 12 till the end.

5 The lclogit command

lclogit is a Stata command that generalizes the routine outlined above. Therefore, this
command does not contain its own ml evaluator and it simply makes use of the Stata
clogit estimation command at each maximization step. This reduces significantly the
estimation time and - importantly - it bases the EM estimation on the quality and the
efficiency of the clogit evaluator.

Following the routine outlined in the previous section, lclogit declares convergence
when the variation in the last 5 values of the maximized log likelihood is smaller than
a given threshold.11 When this happens lclogit stops the internal loop and displays
the estimated coefficients and the class shares.

The results are displayed in a formatted table, with the columns containing the
results by classes.12 When there are more than 5 latent classes the table of results is
divided in blocks of 6 columns. This should allow for a better view of the results and it

11. Users are allowed to change this threshold, the default is 0.001%.
12. lclogit uses the built-in programs estimates store and estimates table in order to display the

table of results. Moreover, the command estimates store allows displaying the last set of internal
clogit estimations by simply typing estimates dir after lclogit.

also avoid visualization problems when the table becomes too wide.13

The syntax for lclogit is:

lclogit depvar
[
varlist

] [
if
][

in
]
, group(varname) id(varname) nclasses(#)[

options
]

The options allowed by lclogit are:

• group(varname) is required; it specifies the variable for the choice occasions.

• id(varname) is required; it specifies the variable for the choice-makers

• nclasses(#) is required; it sets the number of latent classes to be estimated.

• seed(#) is used to set the starting values, the default is 1234567890;14

• niter(#) specifies the number of maximum iterations, the default is 150;

• convergence(#) specifies the minimum variation with respect to the last 5 values
of the log likelihood in order to declare convergence, the default is 0.001%;

• cpname(newvarname), tells lclogit to save C new variables called newvarname1,
newvarname2,.., newvarnameC containing the individual conditional probabilities
of class membership.

6 Application

For our application we use the data illustrated in the previous section and, therefore,
a model without covariates on the class share probabilities. We begin by estimating a
conditional logit model using the Stata command clogit:

. use http://fmwww.bc.edu/repec/bocode/t/traindata.dta, clear

. clogit y price contract local wknown tod seasonal, group(gid)

Iteration 0: log likelihood = -1379.3159

(output omitted )

Iteration 4: log likelihood = -1356.3867

Conditional (fixed-effects) logistic regression Number of obs = 4780
LR chi2(6) = 600.47
Prob > chi2 = 0.0000

Log likelihood = -1356.3867 Pseudo R2 = 0.1812

y Coef. Std. Err. z P>|z| [95% Conf. Interval]

price -.6354853 .0439523 -14.46 0.000 -.7216302 -.5493403

13. For models with more than 20 latent classes lclogit does not show the results in a formatted table
but in a plain matrix, which can be also found in the ereturn list.

14. Note that the seed is set internally only when computing the starting values. Thereafter, the seed
is set back to the original value. This allows for comparable results and simulation-based inference.

contract -.13964 .0161887 -8.63 0.000 -.1713693 -.1079107
local 1.430578 .0963826 14.84 0.000 1.241672 1.619485
wknown 1.054535 .086482 12.19 0.000 .8850338 1.224037

tod -5.698954 .3494016 -16.31 0.000 -6.383769 -5.01414
seasonal -5.899944 .35485 -16.63 0.000 -6.595437 -5.204451

From the results above we can see that - on average - costumers prefer lower prices,
shorter contracts length, a local and well-known company and fixed rate plans.

The Stata command mixlogit from Hole (2007) can be used to estimate a para-
metric mixed logit model with independent, normally-distributed coefficients:15

. mixlogit y, id(pid) group(gid) rand(price contract local wknown tod seasonal)
> nrep(300)

Iteration 0: log likelihood = -1249.8219 (not concave)

(output omitted )

Iteration 7: log likelihood = -1101.6085

Mixed logit model Number of obs = 4780
LR chi2(6) = 509.56

Log likelihood = -1101.6085 Prob > chi2 = 0.0000

y Coef. Std. Err. z P>|z| [95% Conf. Interval]

Mean
price -1.004329 .0721185 -13.93 0.000 -1.145679 -.8629798

contract -.2274985 .047386 -4.80 0.000 -.3203735 -.1346236
local 2.208746 .2439681 9.05 0.000 1.730578 2.686915
wknown 1.656329 .1707167 9.70 0.000 1.32173 1.990927

tod -9.364151 .5858618 -15.98 0.000 -10.51242 -8.215883
seasonal -9.496181 .5792009 -16.40 0.000 -10.63139 -8.360968

SD
price .2151655 .0311095 6.92 0.000 .154192 .2761389

contract .384136 .044778 8.58 0.000 .2963728 .4718992
local 1.788806 .2370063 7.55 0.000 1.324282 2.25333
wknown 1.185838 .1731652 6.85 0.000 .8464401 1.525235

tod 1.6553 .2094545 7.90 0.000 1.244777 2.065824
seasonal -1.119371 .2836182 -3.95 0.000 -1.675252 -.5634893

As it can be seen from the maximized log likelihood, we can reject the conditional
logit specification in favor of a random coefficient model. Moreover, the magnitude of
the coefficients is significantly different when compared to the estimates from clogit.16

We now show how to use lclogit to estimate a nonparametric mixed logit model
by means of the EM algorithm outlined in the previous sections. As explained in section
1, the main idea is to use a latent class model with a relatively high number of classes
so as to approximate he mixing distribution nonparametrically.

15. The Simulated Maximum Likelihood estimation is done with 300 Halton draws; mixlogit can be
installed in Stata by typing findit mixlogit.

16. This is an indication of the bias produced by the IIA property of standard conditional logit models.
See Bhat (2000) for this point.

As stated in Greene and Hensher (2003) and Train (2008), the choice of the
appropriate number of classes is made by means of some information criteria. Here
we opt for the BIC and the CAIC, which penalize more heavily models with a large
number of parameters.17 The next lines show how to use lclogit to estimate a latent
class model with an increasing number of classes:

. forvalues c=2/15 {

. lclogit y price contract local wknown tod seasonal, gr(gid) id(pid) ncl(`c´)

. display e(bic)

. display e(caic)

. display e(ll)

. }

The routine took about 30 minutes to estimate the whole set of 14 models on our
standard-issue PC.18 The next table shows - for an increasing number of latent classes
- the maximized log likelihood, the number of parameters, the BIC and the CAIC:

Classes Log Likelihood N.param. CAIC BIC

Cl.1 -1356.39 6 2746.40 2740.40
Cl.2 -1211.35 13 2495.57 2482.57
Cl.3 -1118.23 20 2348.57 2328.57
Cl.4 -1085.30 27 2321.95 2294.95
Cl.5* -1040.49 34 2271.55* 2237.55
Cl.6 -1028.56 41 2286.93 2245.93
Cl.7 -1006.37 48 2281.79 2233.79
Cl.8* -990.24 55 2288.76 2233.76*
Cl.9 -983.64 62 2314.80 2252.80
Cl.10 -979.23 69 2345.22 2276.22
Cl.11 -965.76 76 2357.52 2281.52
Cl.12 -952.68 83 2370.58 2287.58
Cl.13 -947.24 80 2398.94 2308.94
Cl.14 -945.59 97 2434.89 2337.89
Cl.15 -943.42 104 2469.78 2365.78

From these results we can infer that the 5-class model is optimal according to the
CAIC, whilst the Bayesian criterion points to a model with 8 latent classes.

The output below shows how the program works for a model with 8 latent classes:

. display c(current_time)
12:48:14

. lclogit y price contract local wknown tod seasonal, id(pid) gr(gid) ncl(8)
Iteration 1: log likelihood = -1252.5014
Iteration 2: log likelihood = -1094.1094
Iteration 3: log likelihood = -1043.4077
Iteration 4: log likelihood = -1031.4924
Iteration 5: log likelihood = -1024.0092
Iteration 6: log likelihood = -1016.9219

17. lclogit returns in e() three different information criteria: the AIC, the CAIC and the BIC.
18. We used a PC with a 2.2GHz Intel core 2 duo and 4MB RAM.

Iteration 7: log likelihood = -1008.2313
Iteration 8: log likelihood = -1002.0564
Iteration 9: log likelihood = -998.89545
Iteration 10: log likelihood = -996.875
Iteration 11: log likelihood = -995.84229
Iteration 12: log likelihood = -995.19885
Iteration 13: log likelihood = -994.69055
Iteration 14: log likelihood = -994.31946
Iteration 15: log likelihood = -994.08838

(output omitted )

Iteration 45: log likelihood = -990.23932
Iteration 46: log likelihood = -990.23895
Iteration 47: log likelihood = -990.23877
Iteration 48: log likelihood = -990.23865
Iteration 49: log likelihood = -990.23859
Iteration 151: log likelihood = -990.23853

Latent class model with 8 latent classes

Variable Class1 Class2 Class3 Class4 Class5

price -0.910 -0.737 -0.488 -2.110 -0.642
contract -0.438 0.218 -0.592 -0.662 0.096

local 0.370 2.416 0.782 0.717 2.186
wknown 0.369 2.840 0.710 0.241 1.207

tod -8.257 -6.690 -4.132 -14.191 -3.836
seasonal -6.440 -7.213 -6.560 -17.207 -4.052

Prob 0.120 0.097 0.091 0.070 0.096

Variable Class6 Class7 Class8

price -1.208 -1.533 -0.082
contract -0.198 -0.409 -0.156

local 6.578 0.621 4.937
wknown 5.103 0.930 3.444

tod -14.847 -16.007 -1.088
seasonal -15.334 -14.818 -1.060

Prob 0.111 0.236 0.178

Note: model estimated via EM algorithm

. display c(current_time)
12:49:20

As it can be seen, the EM algorithm took 49 iterations before reaching convergence,
i.e. about one minute in our standard-issue PC. However, the routine is already close
to the maximum at the 11th iteration, i.e. after less than 10 seconds. This is a common
feature of EM algorithms and it actually suggests another useful application of lclogit,
as it could be used to obtain good starting values for the estimation of latent class models
via gradient-based algorithms.19

19. This could be particularly useful either to speed up the estimation process or to avoid convergence
problems when estimating models with a high number of latent classes.

As Train (2008) points out, when the number of classes rises summary statistics for
the distribution of each coefficient could be more informative than the single point esti-
mates. For this reason, lclogit stores the predicted weighted average of each coefficient
in the vector e(PB):

. matrix list e(PB)

e(PB)[1,6]
Average of Average of Average of Average of Average of Average of

price contract local wknown tod seasonal
Coeff -.94577 -.26901 2.3690 1.9184 -9.0036 -9.058

Interestingly, although the parameters are rather different from class to class their
weighted averages are close to the correspondent values obtained from the parametric
estimation via mixlogit.

7 Conclusions

This article has shown how to estimate a nonparametric mixed logit model in Stata
with one of the methods proposed in Train (2008). The method makes use of an EM
algorithm that - thanks to its desirable properties - allows estimating a latent class model
with a high number of classes so that the unobserved distribution of the coefficients can
be approximated nonparametrically. We have also shown how to use the Stata command
lclogit, which performs the EM estimation for latent class logit models automatically.

